

## Aula 09 - Comandos repetitivos.

Outro mecanismo importante na programação e que é fortemente suportado pela Lógica de Programação são as estruturas de **repetições, laços ou "loops"** como também é conhecido.

Este mecanismo está relacionado à necessidade de repetição de um determinado comando ou sequência de comandos, comum a grande parte dos programas computacionais, como por exemplo um menu, ou algum algoritmo que implemente um processo de leitura de vários dados. (SALIBA 1993), (FORBELLONE 2005) (PINTO 1990)

Na lógica de programação, encontramos duas categorias de comandos repetitivos de acordo com a forma com que o laço ou loop acontece:

### Laços contados:

Esta estrutura de repetição possui como característica principal o fato de que se **conhece previamente** quantas vezes os comandos existentes dentro do laço serão executados. Por exemplo, quando um problema coloca de forma explícita quantos dados serão lidos, ou quantas vezes aqueles comandos serão executados, etc

Observe neste caso, problemas da forma: "*considere a leitura de n dados para a..*", "*existem n animais no pasto*", "*Seja dado n números..*".

O comando em algoritmo que implementa esta estrutura é o comando **for..next**.

O comando **for..next** é aquele que permite a repetição de uma instrução simples ou estruturada da linguagem de programação, sendo que o número de vezes considerado está relacionado à uma variável **contador (ou índice)** que conta com intervalo específico de um-em-um (o intervalo é chamado de **steps** e o algoritmo implementa steps de um-em-um, enquanto outras linguagens podem utilizar de dois em dois, três em três, etc).

Na sua forma genérica, é assim especificada a estrutura **for..next**:

```
for <variável> = <valor-inicial> to <valor-limite> [step <valor a incrementar>]
    <seqüência-de-comandos>
[for exit]
    <seqüência-de-comandos>
next <variável>
```

Sintaxe:

Neste caso,

- **valor-inicial** e **valor-final** são constantes, variáveis ou expressões cujos valores estejam contidos num conjunto de inteiros ou caractere.
- O elemento **<variável>** é uma variável (denominada **índice**) com valores pertencentes ao mesmo conjunto formado pelos valores de **<valor-inicial>** e **<valor-limite>**.

- O **corpo** do loop corresponde a uma instrução simples ou uma sequência de comandos (<seqüência-de-comandos>)
- O comando opcional [**step** <numero>] determina qual o valor do passo, ou seja, a quantidade em que o contador é alterado cada vez que passa pelo loop. Se o passo não for especificado, o valor default será 1.

Considere o seguinte exemplo:

```
Sub exemplo_01()
Dim i As Integer
For i = 1 To 10 Step 1
    Cells(1, i).Value = i + 1
Next i
End Sub
```

O laço funciona da seguinte forma:

- ☐ A variável **i** assume o valor inicial de 1;
- ☐ Valor de **i** é comparado ao valor final de 10;
- ☐ Se **i <= 10** então as instruções do laço são executados e o valor de **i** passa a assumir o próximo valor no conjunto (ou seja, por exemplo, de i=1, ele passa a valer 2) somando o step que neste caso é 1.
- ☐ Após a mudança no valor de i, este será novamente comparado com o **valor 10**
- ☐ Caso o conjunto discreto seja o dos inteiros, o valor de **i** é apenas incrementado; mas para outros conjuntos discretos (enumerados, por exemplo), **i** vai assumir efetivamente o valor do próximo elemento do conjunto definido.
- ☐ quando **i > 10**, então o laço é dado por terminado.
- ☐ O resultado da execução deste comando

	A	B	C	D	E	F	G	H	I	J	K	L
1	2	3	4	5	6	7	8	9	10	11		
2												
3												

Considere o outro exemplo:

```
Sub Beeps()
Dim x As Integer
For x = 1 To 50
    Beep
Next x
End Sub
```

Ao executar essa macro, um beep será ouvido pelo usuário.

Considere esse outro exemplo:

```

Sub exemplo_03()
Dim i As Integer, j As Integer

For i = 1 To 8
    For j = 1 To 3
        Cells(i, j).Value = 4
    Next j
Next i
End Sub

```

Este exemplo mostra a utilização de dois comandos **for** sendo um mais interno que o outro. Neste caso, para cada valor da variável **i** temos um conjunto de valores **j** que são executados. Assim:

Quando **i** = 1, **j** vai variar de 1..3

Quando **i** = 2, **j** vai variar de 1..3

Quando **i** = 3, **j** vai variar de 1..3

O resultado da execução desta macro é:

	A	B	C	D
1	4	4	4	
2	4	4	4	
3	4	4	4	
4	4	4	4	
5	4	4	4	
6	4	4	4	
7	4	4	4	
8	4	4	4	
9				
...				

Considere o exemplo abaixo:


```

Sub exemplo_04()
Dim i As Integer, total As Integer, s As String
s = "" ' inicialização da variável s
For i = 16 To 2 Step -2
    s = s & vbCrLf & " o valor i = " & i & " e o valor de total = " & total
    total = total + i

Next i
MsgBox s & vbCrLf & " O valor de total = " & total
End Sub

```

Observe que os valores de **i** são decrementados de dois. O resultado é mostrado abaixo:



## Uso de acumuladores e contadores

O uso de **contadores e acumuladores** é muito comum em linguagem de programação e seu conceito deve ser muito bem entendido.

Geralmente, utilizamos um contador quando desejamos realizar a contagem de algum item e um acumulador quando necessitamos acumular numa variável um determinado conjunto de valores.

### Exemplo de (contador):

Suponha seja dado um certo conjunto de letras do alfabeto lidos através de uma frase e desejamos saber quantas de cada vogal existem no conjunto, assim como espaços em branco e consoantes.

Assim precisamos de um mecanismo para ler os dados, depois identificar as vogais e realizar a contagem de termos. Para realizar esta tarefa, criamos um contador para cada vogal e inicializamos em zero antes de inicializar o comando para.

```
Sub exemplo_05()  
Dim cont_a As Integer, cont_e As Integer, cont_i As Integer, i As Integer  
Dim frase As String, cont_o As Integer, cont_u As Integer  
Dim cont_branco As Integer, cont_consoante As Integer, s As String  
  
frase = InputBox(" Entre com uma frase ")  
For i = 1 To Len(frase) ' tamanho da frase  
  
If UCase(Mid(frase, i, 1)) = "A" Then ' verifica se é A  
    cont_a = cont_a + 1  
ElseIf UCase(Mid(frase, i, 1)) = "E" Then ' verifica se é E  
    cont_e = cont_e + 1  
ElseIf UCase(Mid(frase, i, 1)) = "I" Then ' verifica se é I  
    cont_i = cont_i + 1  
ElseIf UCase(Mid(frase, i, 1)) = "O" Then ' verifica se é O  
    cont_o = cont_o + 1  
ElseIf UCase(Mid(frase, i, 1)) = "U" Then ' verifica se é U  
    cont_u = cont_u + 1  
ElseIf UCase(Mid(frase, i, 1)) = " " Then ' verifica se é branco  
    cont_branco = cont_branco + 1  
Else  
    cont_consoante = cont_consoante + 1  
  
End If  
  
Next i  
s = " A frase = " + frase  
s = s & vbCrLf & " quantidade de letras A = " & cont_a
```

```

s = s & vbCrLf & " quantidade de letras E = " & cont_e
s = s & vbCrLf & " quantidade de letras I = " & cont_i
s = s & vbCrLf & " quantidade de letras O = " & cont_o
s = s & vbCrLf & " quantidade de letras U = " & cont_u
s = s & vbCrLf & " quantidade de brancos = " & cont_branco
s = s & vbCrLf & " quantidade de consoantes = " & cont_consoante


```

```


MsgBox s
End Sub

```

Considere a seguinte entrada de dados:



Um exemplo de execução deste algoritmo pode ser mostrado abaixo:



O exemplo acima mostra que dentro de um laço as variáveis cont\_a, cont\_e, cont\_i, cont\_o, cont\_u, cont\_branco e cont\_consoante são incrementadas de 1 em cada passo do laço caracterizando assim a presença de um contador.

#### Observação:

- A **função Len(string)** - Esta função determina o tamanho da String que foi passada como parâmetro para a função
- A **Função Ucase(string)** – Esta função converte para MAIÚSCULAS, a String que foi passada como parâmetro para a função
- A **Função LCase(string)** - Esta função converte para MINUSCULAS a String que foi passada como parâmetro para a função
- A **função MID**. Esta função retorna um número especificado de caracteres, a partir de uma posição especificada, dentro da String. Os espaços em branco também devem ser considerados.

O formato geral da função é o seguinte:

**Mid(String, posicao\_inicio, n)**

onde:


- posicao\_inicio** - é a posição a partir da qual devem ser retornados caracteres

- **n** - é o número de caracteres a retornar.

Considere alguns exemplos:

```
Sub exemplo10()  
MsgBox Mid("Jose da Silva Mestre", 8, 6)  
End Sub
```

Teremos como resultado



Este exemplo de uso da função, retorna, a partir da posição 8, 6 caracteres, neste caso, retornará "Silva".

Desta forma, o comando **UCase(Mid(frase, i, 1)) = "A"** significa que i estará variando de posição inicial até a posição final, de tal forma que em cada iteração, obterá somente um caractere. Este será comparado com o caractere entre aspas. Se a comparação for verdadeira, será acrescentado um (1) no contador.

#### Exemplo de (acumulador):

Sejam dados **n** notas de alunos e desejamos saber a média de todos eles. Neste caso, devemos somar a nota de todos os alunos e dividir pelo número de alunos, que no caso é **n**.

O fato de estarmos trabalhando com uma variável que deverá acumular todas as notas dos alunos temos que inicializar a variável em zero (0) e depois em cada passo da leitura de novas notas acumular os valores numa única variável chamado de acumulador, pois realiza a atividade de acumular conteúdo que seja diferente de um.

Neste sentido, serão lidas as quantidades de notas, depois cada uma das notas e ao final, calcular a média aritmética de todas as notas. A média é calculada dividindo-se a soma de todas as notas pelo total de notas.

Para resolver este problema vamos utilizar um formulário que contenha notas de duas disciplinas e depois utilizamos um outro botão pra realizar o cálculo da média do aluno.

Assim temos:

	A	B	C	D
1	Nome do aluno	Nota matemática	Nota portugues	
2	José da Silva	8,5	5,2	
3	Pedro carlos	9,4	4,3	
4	Antonio Martins	4,7	7,8	
5	Carla deAndrade	3,8	6,2	
6	Pricila Campos	9,7	8,4	
7				
8				

#### Exemplo de codigo

```

Sub repetitivo01()
'declaração de variaveis
Dim soma1 As Double, cont1 As Integer
Dim soma2 As Double, cont2 As Integer
Dim linha As Integer, s As String
Dim i As Integer
'verificar quantas linhas ocupadas possui a planilha
linha = Range("a300").End(xlUp).Row
'inicializar as variaveis
soma1 = 0 ' acumulador
soma2 = 0 ' acumulador
cont1 = 0 ' contador
cont2 = 0 ' contador
For i = 2 To linha
' notas de matematica
soma1 = soma1 + Cells(i, 2).Value 'coluna 2
cont1 = cont1 + 1
' notas de portugues
soma2 = soma2 + Cells(i, 3).Value 'coluna 3
cont2 = cont2 + 1
Next i
'soma das notas de matematica
s = " Soma das notas de matemática = " & soma1
s = s & vbCrLf & " Soma das notas de portugues " & soma2
'mostrar resultados
MsgBox s
End Sub

```

Considere um outro exemplo, onde serão calculos media dos alunos em recuperação, ou seja entre 4,0 e 6,0

```

Sub exercicio01()
'declaração de variaveis
Dim soma As Double, s As String
Dim i As Integer, linha As Integer, media As Double
Dim n1 As Double, soma1 As Double, cont1 As Integer
'calculos e operações
linha = Range("A300").End(xlUp).Row
soma = 0
soma1 = 0
cont1 = 0
For i = 2 To linha
n1 = Cells(i, 2).Value 'armazena nota n variavel n1
soma = soma + n1 'acumula nota
If n1 >= 6 And n1 <= 10 Then
soma1 = soma1 + n1 ' soma as notas
cont1 = cont1 + 1 'conta quantas notas
End If
Next i
'observe que o comando condicional abaixo
'será necessário, pois não existe divisão por zero
If cont1 <> 0 Then ' teste se contador <> 0
media = soma1 / cont1 'calcula media
Else: media = 0 'quando o contador é zero
End If
s = " A soma das notas de matemática = " & soma
s = s & vbCrLf & " A media dos aprovados = " & media
MsgBox s
End Sub

```

#### Exercicios:

- Modifique o codigo acima para que ele possa mostrar a média aritmetica de cada uma das notas, ou seja, a media das notas de matemática e portugues
- Calcule a media das notas acima de 6,0 em matematica
- Calcule a media das notas acima de 6,0 em portugues
- Um aluno em recuperação possui nota entre 4,0 e 6,0. Conte quantos alunos são em cada uma das disciplinas

## Cálculo do Maior/Menor

Um outro exercício muito comum dentro do conceito de comandos repetitivos utilizando-se planilha com dados é o cálculo do maior ou menor nota, ou maior/menor número de um certo conjunto de dados

O algoritmo que resolve este problema tem os seguintes passos:

- 1) primeiro será a inicialização da variável que irá armazenar o menor/maior elemento
- 2) Através da iteração dentro de um laço For comparar o conteúdo do maior/menor com a nota obtida pela planilha de dados
- 3) Quando essa comparação for verdadeira, armazenar esse novo valor na variável maior/menor

Para o exemplo anterior observe um código considerando que estamos sempre inicializando a variável maior/menor com o primeiro elemento que aparece na planilha de dados, ou seja, linha 2. (observe que na linha 1 temos os títulos dos dados constante em cada coluna)

Exemplo de código VBA utilizando uma planilha de dados que tenha nome do aluno, nota de português e nota de matemática

```
Sub achar_maior()  
'declaração de variaveis  
Dim maior As Double, linha As Integer, i As Integer, aux As Integer  
Dim n1 As Double, n2 As Double, aux2 As Integer  
Dim s As String, menor As Double  
'calculos e operações  
linha = Range("A300").End(xlUp).Row  
maior = Cells(2, 2).Value ' inicializar nota matematica  
menor = Cells(2, 3).Value ' inicializa nota de portugues  
For i = 2 To linha  
    n1 = Cells(i, 2).Value 'nota de Matemática  
    n2 = Cells(i, 3).Value 'nota de portugues  
    If n1 >= maior Then  
        maior = n1 'guarda maior nota de matematica  
        aux = i ' guarda a linha da maior nota  
    End If  
    If n2 <= menor Then  
        menor = n2 ' guarda menor nota de portugues  
        aux2 = i 'guarda a linha da menor nota  
    End If  
Next i  
'mostrar dados  
s = " O valor da maior nota de matematica = " & maior  
s = s & vbCrLf & "O nome do aluno da maior nota de matematica " & Cells(aux, 1).Value  
s = s & vbCrLf & "A menor nota de Portugues = " & menor  
s = s & vbCrLf & "O nome do aluno da menor nota Portugues = " & Cells(aux2, 1).Value  
MsgBox s  
End Sub
```

Que gera a seguinte saída de dados:



O valor da maior nota de matematica = 9,7  
O nome do aluno da maior nota de matematica Paulo  
A menor nota de Portugues = 4,3  
O nome do aluno da menor nota Portugues = Pedro

OK

**Exercícios:**

**Resolvido 03** – Faça um programa que leia um numero N que indica quantos valores inteiros e positivos devem ser lidos a seguir. Para cada numero lido, mostre uma tabela contendo o valor lido e o fatorial desse numero

**Resolvido 04** – Foi feita uma estatística em cinco cidades brasileiras para coletar dados sobre acidentes de trânsito.

Foram obtidos os seguintes dados:

- a) código da cidade;
- b) número de veículos de passeio;
- c) número de acidentes de trânsito com vítimas.

Deseja-se saber:

- a) qual é o maior e qual é o menor índice de acidentes de trânsito e a que cidades pertencem;
- b) qual é a média de veículos nas cinco cidades juntas;
- c) qual é a média de acidentes de trânsito nas cidades com menos de 2.000 veículos de passeio

**Resolvido 07** - Faça um programa que monte dado um valor N inteiro calcule o seu numero fibonacci. Mostre a sequencia ate este numero.

0 - 1 - 1 - 2 - 3 - 5 - 8 - 13 - 21 - 55...

Fib(0) = 0  
Fib(1) = 1  
Fib(2) = 1  
Fib(3) = 2  
Fib(4) = 3  
Fib(5)=5  
Fib(6)=8  
Fib(7)=13  
Fib(8)=21  
Fib(9)=34  
Fib(10) = 55

**Resolvido 11** – Faça um programa que receba um numero inteiro maior que 1. Verifique se o numero fornecido é primo ou não e mostre uma mensagem de número primo ou de número não primo.  
Um número é primo quando é divisível apenas por 1 ou por ele mesmo.

**Resolvido 12** – Em uma fábrica trabalham homens e mulheres divididos em três classes:

- trabalhadores que fazem até 30 peças por mês — classe 1;
- trabalhadores que fazem de 31 a 50 peças por mês — classe 2;
- trabalhadores que fazem mais de 50 peças por mês — classe 3.

A classe 1 recebe salário mínimo. A classe 2 recebe salário mínimo mais 3% deste salário por peça, acima das 30 peças iniciais. A classe 3 recebe salário mínimo mais 5% desse salário por peça, acima das 30 peças iniciais.

Faça um programa que receba o número do operário, o número de peças fabricadas no mês, o sexo do operário, e que também calcule e mostre:

- o número do operário e seu salário;
- o total da folha de pagamento da fábrica;
- o número total de peças fabricadas no mês;
- a média de peças fabricadas pelos homens;
- a média de peças fabricadas pelas mulheres; e
- o número do operário ou operária de maior salário.

A fábrica possui 15 operários.

#### **Referências:**

FORBELLONE, A. L. V.- **Lógica de Programação: a Construção de Algoritmos e Estruturas de Dados**. SÃO PAULO, MAKRON BOOKS/PEARSON PRENTICE HALL, 3ªed 2005

SALIBA, W. L. C. - **Técnicas De Programação: uma Abordagem Estruturada**. SÃO PAULO, MAKRON, MCGRAW-HILL, 1993

PINTO, W. S. **Introdução ao Desenvolvimento de Algoritmos e Estruturas de Dados** – SÃO PAULO:EDITORA ÉRICA. 1990